



```
<relational-expression> →
    <additive-expression>
    | <relational-expression> < <additive-expression>
    | <relational-expression> > <additive-expression>
    | <relational-expression> <= <additive-expression>
    | <relational-expression> >= <additive-expression>

<additive-expression> →
    <multiplicative-expression>
    | <additive-expression> + <multiplicative-expression>
    | <additive-expression> - <multiplicative-expression>

<multiplicative-expression> →
    <unary-expression>
    | <multiplicative-expression> * <unary-expression>
    | <multiplicative-expression> / <unary-expression>

<unary-expression> →
    <primary-expression>
    | <identifier> ( <argument-expression-list> )

<primary-expression> →
    <identifier>
    | <constant>
    | ( <expression> )

<argument-expression-list> →
    <expression>
    | <argument-expression-list> , <expression>

<constant> →
    <integer-constant>
    | <character-constant>

<integer-constant> →
    <digit>
    | <integer-constant> <digit>

<character-constant> → ' <letter> '

<identifier> →
    <letter>
    | <identifier> <letter>
    | <identifier> <digit>

<letter> →
    a | b | c | d | e | f | g | h | i | j | k | l | m |
    n | o | p | q | r | s | t | u | v | w | x | y | z |
    A | B | C | D | E | F | G | H | I | J | K | L | M |
    N | O | P | Q | R | S | T | U | V | W | X | Y | Z

<digit> →
    0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9
```

Figure 7.8

(Continued)

